METR 4433 – Mesoscale Meteorology
Spring 2005

Problem Set #2

Example Answers

1. (50%) A vertical wind profile is given by the following table:

z (height, km)	θ (direction, deg)	V(speed, m/s)
0	110	6
1	150	10
2	180	15
3	190	17
4	250	25
5	270	30
6	310	40

Assume that the storm motion vector is from 225 degrees (from SW) and the speed is 12 m/s.

a). Plot the hodograph and the storm-relative velocity vectors at each level

b). Calculate the horizontal vorticity (vector, in terms of the vorticity components or in magnitude and direction) in each of the six layers between the levels of observations

c). Determine the mean (storm-relative) wind vector in each of these size layers

d). Using the layer-mean wind obtained above, calculate the storm-relative helicity in each of the six layers, and determine the vertically integrated environmental helicity in the lowest three kilometers

e) Calculate the (storm-relative) streamwise vorticity and (storm-relative) relative helicity in each of the six layers

f). Discuss your results and their significance in terms of the their effect on the behavior and type of the storms that occur in such environment

g). For this wind profile, what kind of CAPE values will give you a BRN that suggests a high probability of multicell and supercell storms, respectively?

The following is a Fortran 90 program to calculate the above quantities, and to produce the hodograph.

```
PROGRAM HELICITY
!
! Program to calculate parameters of Mesoscale Meteorology Homework #2
! It calls ZXPLOT (http://www.caps.ou.edu/ZXPLOT) graphics library to
! for plotting.
! Written by Ming Xue
! 5/2005
!
  IMPLICIT NONE

  INTEGER, PARAMETER :: n = 6
  REAL :: speed(0:n),direction(0:n),z(0:n)
  REAL :: ua(0:n), va(0:n), ur(0:n), vr(0:n)
                                 ! u-v components of abs and rel. velocity

  REAL :: ustorm, vstorm, Cstorm,dirstorm ! storm motion vector
  REAL :: omegax(n),omegay(n) ! horizontal vorticity
  REAL :: H(n)     ! Storm-relative helicity
  REAL :: um(n), vm(n)   ! layer mean velocity
  REAL :: h3km
  REAL :: omegas(n) ! streamwise vorticity components
  REAL :: RH(n)     ! relative helicity
  REAL CAPE, BRN, S, u6km,v6km,u500m,v500m
  INTEGER i
  REAL :: alpha, pi, deg2rad

  data z/ 0.0, 1000.0, 2000.0, 3000.0, 4000.0, 5000.0, 6000.0/
  data speed/6.0,10.0,15.0,17.0,25.0,30.0,40.0/
  data direction/110.0,150.0,180.0,190.0,250.0,270.0,310.0/

  CALL XDEVIC
  CALL XPSPAC(0.1,0.9, 0.1, 0.9)
  CALL XMAP (-40.0, 40.0, -40.0, 40.0)
  CALL XAXFMT('(I3)')
  CALL XAXES(0.0, 5.0, 0.0, 5.0)


  CALL Xdash
  DO i=5,40,5
    CALL Circle(0.0,0.0,float(i))
  ENDDO
  CALL xfull

  deg2rad = atan(1.0)/45.0

  cstorm = 12.0
  dirstorm = 225.0

  alpha = 360.0-(dirstorm-180.0-90.0)
  ustorm = cstorm*cos(alpha*deg2rad)
  vstorm = cstorm*sin(alpha*deg2rad)

  CALL XPENUP(0.0, 0.0)
  CALL XPENDN(ustorm, vstorm)
!
! Determine Cartesian components of the absolute velocity
! at each observation level
!
  DO i=0,n
    alpha = 360.0-(direction(i)-180.0-90.0)
             ! Convert to polar angle from x-axis in counterclock wise
direction
```

```
    ua(i) = speed(i) * cos(alpha*deg2rad)
    va(i) = speed(i) * sin(alpha*deg2rad)
    CALL XPENUP(0.0, 0.0)
    CALL XPENDN(ua(i),va(i))
  ENDDO

  CALL XPENUP(ua(0),va(0))
  DO i=1,n
    CALL XPENDN(ua(i),va(i))
  ENDDO
!
! Detemine storm-relative velocity at observation level
!
  DO i=0,n
    ur(i) =ua(i)-ustorm
    vr(i) =va(i)-vstorm
    CALL XPENUP(ustorm,vstorm)
    CALL XPENDN(ua(i),va(i))
  ENDDO
!
! Calculate horizontal vorticity in each layer
!
  DO i=1,n
    omegax(i) = -(va(i)-va(i-1))/(z(i)-z(i-1))
    omegay(i) = +(ua(i)-ua(i-1))/(z(i)-z(i-1))
  ENDDO
!
! Calculate mean storm-relative velocity in each layer
!
  DO i=1,n
    um(i) = 0.5*(ur(i)+ur(i-1))
    vm(i) = 0.5*(vr(i)+vr(i-1))
    CALL XPENUP(ustorm,vstorm)
    CALL XPENDN(ustorm+um(i),vstorm+vm(i))
  ENDDO
!
! Calculate storm-relative helicity in each layer
! H = V dot H, where V is the layer mean velocity
!
  DO i=1,n
    H(i) = um(i)*omegax(i)+vm(i)*omegay(i)
  ENDDO
!
! Vertically integrated storm-relative helicity in lowest 3km
!
  h3km = 0.0
  DO i=1,3
    h3km = h3km + h(i)*(z(i)-z(i-1))
  ENDDO
!
! Calculate streamwise vorticity and relative helicity in each layer
!
  DO i=1,n
    omegas(i) = h(i)/sqrt(um(i)**2+vm(i)**2)
    RH(i) = omegas(i)/sqrt(omegax(i)**2+omegay(i)**2)
  ENDDO

!
! Determine mean wind in the 6 km layer for calculating BRN
! We will use storm relative velocity here. Using abs. velocity
! should give the same V6km - V500m.
!
  u6km = 0.0
```

```
  v6km = 0.0
  DO i=1,n
    u6km=u6km+um(i)
    v6km=v6km+vm(i)
  ENDDO
  u6km = u6km/n
  v6km = v6km/n
!
! Determine mean wind in the lowest 500 m for calculating BRN
!
  u500m = 0.5*(ur(0)+ur(1))  ! find linearly interpolated value at 500m
  v500m = 0.5*(vr(0)+vr(1))  ! find linearly interpolated value at 500m

  u500m = 0.5*(ur(0)+u500m)  ! now mean velocity in the first 500m
  v500m = 0.5*(vr(0)+v500m)  ! now mean velocity in the first 500m

  S = sqrt( (u6km-u500m)**2+(v6km-v500m)**2 ) ! BRN shear

  WRITE(6,'(/10a)')  &
  '     Z(km)   Dir(deg)   V(m/s)    ua(m/s)   va(m/s)    ur(m/s)  vr(i)'
  DO i=0,n
  WRITE(6,'(7f10.3)')z(i)*0.001,direction(i),speed(i),ua(i),va(i),ur(i),vr(i)
  ENDDO

  WRITE(6,'(/10a)') &
  '   Layer omegax(1/s) omegay(1/s)  um(m/s)  vm(m/s)    H(m/s**2)  RH
OmegaS'
  DO i=1,n
  WRITE(6,'(i7,2F11.6,4f10.3,f11.6)') &
  i,omegax(i),omegay(i),um(i),vm(i),h(i),rh(i),omegas(i)
  ENDDO

  WRITE(6,'(/3x,a,f10.3)') '3km integrated helicity (m**2/s**2)=', h3km
  WRITE(6,'(3x,a,f10.3)') 'BRN shear S (m/s) =', S

  WRITE(6,'(3x,a,f10.3)') 'CAPE for BRN=10 is ', 10*0.5*S**2
  WRITE(6,'(3x,a,f10.3)') 'CAPE for BRN=45 is ', 45*0.5*S**2

  CALL XGREND

  STOP
END PROGRAM HELICITY

SUBROUTINE CIRCLE(x0, y0, r)
  REAL :: x0, y0, r

  deg2rad = atan(1.0)/45.0

  x=x0+r*cos(0*deg2rad)
  y=y0+r*sin(0*deg2rad)
  call xpenup(x,y)

  DO i=1,360
    x=x0+r*cos(i*deg2rad)
    y=y0+r*sin(i*deg2rad)
    call xpendn(x,y)
  ENDDO
END SUBROUTINE CIRCLE
```

Output of the program:

Z(km)	Dir(deg)	V(m/s)	ua(m/s)	va(m/s)	ur(m/s)	vr(i)
0.000	110.000	6.000	-5.638	2.052	-14.123	-6.433
1.000	150.000	10.000	-5.000	8.660	-13.485	0.175
2.000	180.000	15.000	0.000	15.000	-8.485	6.515
3.000	190.000	17.000	2.952	16.742	-5.533	8.256
4.000	250.000	25.000	23.492	8.551	15.007	0.065
5.000	270.000	30.000	30.000	0.000	21.515	-8.485
6.000	310.000	40.000	30.642	-25.711	22.157	-34.197

Layer	omegax(1/s)	omegay(1/s)	um(m/s)	vm(m/s)	H(m/s**2)	RH	OmegaS
1	-0.66E-02	0.64E-03	-13.804	-3.129	0.089	0.949	0.63E-02
2	-0.63E-02	0.50E-02	-10.985	3.345	0.086	0.932	0.75E-02
3	-0.17E-02	0.30E-02	-7.009	7.386	0.034	0.975	0.33E-02
4	0.82E-02	0.21E-01	4.737	4.161	0.124	0.891	0.20E-01
5	0.86E-02	0.65E-02	18.261	-4.210	0.129	0.639	0.69E-02
6	0.26E-01	0.64E-03	21.836	-21.341	0.548	0.697	0.18E-01

```
3km integrated helicity (m**2/s**2)=   209.603
BRN shear S (m/s) =    16.326
CAPE for BRN=10 is   1332.728
CAPE for BRN=45 is   5997.276
```

Hodograph:



g). For this wind profile, what kind of CAPE values will give you a BRN that suggests a high probability of multicell and supercell storms, respectively?

Multicell storms occur mostly when bulk Richardson number Rn = 2* CAPE/ $(V_{6km}-V_{BL})^2 > 45$. See notes.

2. (50%) The storm-relative environment helicity in the lowest 3 km layer is given as

$$SREH = \int_{0km}^{3km} [(\vec{V} - \vec{C}) \cdot \vec{\omega}_H] dz$$

which, based on definition $\vec{\omega}_H = \hat{k} \times \frac{d\vec{V}}{dz} = -\frac{dv}{dz}\hat{i} + \frac{du}{dz}\hat{j}$, can be rewritten as

$$SREH = -\int_{0km}^{3km} \hat{k} \cdot \left[(\vec{V} - \vec{C}) \times \frac{d\vec{V}}{dz} \right] dz = -\int_{0km}^{3km} \hat{k} \cdot \left[(\vec{V} - \vec{C}) \times d\vec{V} \right] = -\int_{0km}^{3km} \hat{k} \cdot \left[\vec{V}_r \times d\vec{V} \right]$$

where $\vec{V}_r \equiv (\vec{V} - \vec{C})$ is the storm-relative velocity.

a). Using the above information and your knowledge of analytic geometry, show that the SREH is equal to minus twice the signed (i.e., positive or negative) area swept out by the storm-relative wind vector between 0 and 3 km on a hodograph. Note that, by convention, an area is positive (negative) if it is swept by a vector in a counterclockwise (clockwise) direction.

$$SREH = -\int_{0km}^{3km} \hat{k} \cdot \left[\vec{V}_r \times d\vec{V} \right]$$

There is more than one way for showing this. The following is only one example.

Consider the SERH in a layer of depth *dz* in which the wind vector increases from $\vec{V}_r$ to $\vec{V}_r + d\vec{V}$, the SERH in the layer is

$$d(SREH) = -\hat{k} \cdot \left[\vec{V}_r \times d\vec{V} \right]$$

The total SERH in the 3 km layer will be sum of the SERH in each of such layers.



In the above Figure, we can see that

$$\begin{aligned} d(SREH) &= -\hat{k}\cdot\left[\vec{V}_r \times d\vec{V}\right] \\ &= -\hat{k}\cdot(-\hat{k})\left|\vec{V}_r\right|\left|d\vec{V}\right|\sin\theta = \left|\vec{V}_r\right|\left|d\vec{V}\right|\sin(\pi-\theta) \\ &= \left|\vec{V}_r\right|\left|d\vec{V}\right|\sin(\pi-\theta) = \left|\vec{V}_r\right|\left|d\vec{V}\right|\sin\theta' = \left|\vec{V}_r\right| h = -2*Area \end{aligned}$$

where $Area = -\frac{\left|\vec{V}_r\right| h}{2}$. The negative sign is because of the definition of the area, which in this case is swept by $\vec{V}_r$ in the clockwise direction ($d\vec{V}$ points to the right side of $\vec{V}_r$).

$$\begin{aligned} \therefore SREH &= -\int_{0km}^{3km} \hat{k}\cdot\left[\vec{V}_r \times d\vec{V}\right] = \int_{0km}^{3km} d(SREH) \\ &= -2\times Total\_Area\_swepted\_by\_\vec{V}_r\_in\_3km\_depth \end{aligned}$$

To keep the problem simple, let's assume that wind observations are available at the 0 and 3 km levels only (note – the above solution consider the general case where $\vec{V}_r$ varies continuously with height).

b). If the storm-relative velocities at 0 and 3 km levels are (u_{r1}, v_{r1}) *and* (u_{r2}, v_{r2}), respectively, show that SREH can be calculated from

$$SERH = u_{r2}v_{r1} - u_{r1}v_{r2}\,.$$

Hint: Think of how you calculated the storm-relative helicity in Problem 1, for each of those 6 layers. Also, $d\vec{V} = d\vec{V}_r$ because the storm motion vector is constant with height.

The SERH in the 3 km layer is:

$$SREH = [(\vec{V} - \vec{C}) \cdot \vec{\omega}_H]dz = (\overline{u}\hat{i} + \overline{v}\hat{j}) \cdot \left[-\frac{dv}{dz}\hat{i} + \frac{du}{dz}\hat{j} \right] dz = (\overline{u}\hat{i} + \overline{v}\hat{j}) \cdot (-dv\hat{i} + du\hat{j})$$

$$= -\frac{u_1 + u_2}{2}(v_2 - v_1) + \frac{v_1 + v_2}{2}(u_2 - u_1)$$

$$= -\frac{1}{2}\left[u_1v_2 - u_1v_1 + u_2v_2 - u_2v_1 - v_1u_2 + v_1u_1 - v_2u_2 + v_2u_1\right]$$

$$= u_2v_1 - u_1v_2$$

c). Verify that for the following hodograph and a zero storm-motion vector, the above two methods give the same results.



Solution: Just plug in the numbers.

d) Explain why larger SERH tends to promote longer lasting supercell storms?

See notes. The key is that it leads to large correlation between w' and ζ', implying large vorticity in updraft – from our analysis on pressure perturbation associated with rotation updraft, we understand rotation in updraft produces additional positive lifting therefore stronger updraft therefore stronger storms.